

## Example 2b: Practice with Interactions among Continuous Predictors in General Linear Models (as estimated using restricted maximum likelihood in SAS MIXED)

The models for this example come from Hoffman (2015) chapter 2. We will be examining the extent to which cognition (as measured by an information test outcome) can be predicted from age (centered at 85 years) grip strength (centered at 9 pounds), sex (with men as the reference group) and dementia status (none, future, current, with none as the reference) in a sample of 550 older adults. This model follow the main effects model given in example 2a; now we examine interactions involving age, grip strength, and sex (a binary predictor we will treat as continuous in the models that follow for now).

### MIXED Syntax and Output for Equation 2.9, adding interaction of age85\*grip9:

**SAS syntax to create and merge fake people (for getting predicted outcomes) into current data:**

```
* Creating 'fake people' to show age*grip interaction;
* Each row is a fake person for which to create a predicted outcome;
DATA work.FakeAgeGrip; INPUT personID grip9 age85 sexMW demNF demNC;
DATALINES;
-99 3 -5 0 0 0
-99 3 0 0 0 0
-99 3 5 0 0 0
-99 0 -5 0 0 0
-99 0 0 0 0 0
-99 0 5 0 0 0
-99 -3 -5 0 0 0
-99 -3 0 0 0 0
-99 -3 5 0 0 0
; RUN;
* Merge with real data;
DATA work.PlotAgeGrip; MERGE work.Chapter2 work.FakeAgeGrip; BY personID; RUN;
```

Note that the variables that are not part of the age\*grip interaction are held constant at 0, but they must be included for predicted outcomes to be created.

The dataset "PlotAgeGrip" will be used in estimating this model instead of the original Chapter2 dataset.

$$\text{Cognition}_i = \beta_0 + \beta_1 (\text{Age}_i - 85) + \beta_2 (\text{Grip}_i - 9) + \beta_3 (\text{SexMW}_i) + \beta_4 (\text{DemNF}_i) + \beta_5 (\text{DemNC}_i) + \beta_6 (\text{Age}_i - 85)(\text{Grip}_i - 9) + e_i$$

### SAS Model Syntax and Output:

```
TITLE1 'Eq 2.9: Adding Age by Grip Interaction';
TITLE2 'Age 0=85 Grip 0=9';
* Estimate model on data with fake people to make predictions;
PROC MIXED DATA=work.PlotAgeGrip COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
MODEL cognition = age85 grip9 sexMW demNF demNC age85*grip9
/ SOLUTION DDFM=BW COVB OUTPM=PredAgeGrip;
ODS OUTPUT CovParms=CovAgeGrip SolutionF=FixAgeGrip COVB=CovBAgeGrip;
```

On MODEL after /: COVB adds the asymptotic covariance matrix for regions of significance macro; OUTPM = saves predicted outcomes in new named dataset.

We can use the model equation to calculate the **simple effect of age** at any level of *grip strength* (age as the effect, grip strength as the moderator):

$$\text{Simple Effect of Age} = \beta_1(\text{Age}_i - 85) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) \rightarrow (\text{Age}_i - 85)[\beta_1 + \beta_6(\text{Grip}_i - 9)]$$

```
ESTIMATE 'Age Slope at Grip Strength = 6' age85 1 age85*grip9 -3;
ESTIMATE 'Age Slope at Grip Strength = 9' age85 1 age85*grip9 0;
ESTIMATE 'Age Slope at Grip Strength = 12' age85 1 age85*grip9 3;
```

Likewise, we can use the model equation to calculate the **simple effect of grip strength** at any level of *age* (grip strength as the effect, age as the moderator):

$$\text{Simple Effect of Grip} = \beta_2(\text{Grip}_i - 9) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) \rightarrow (\text{Grip}_i - 9)[\beta_2 + \beta_6(\text{Age}_i - 85)]$$

```
ESTIMATE 'Grip Strength Slope at Age = 80' grip9 1 age85*grip9 -5;
ESTIMATE 'Grip Strength Slope at Age = 85' grip9 1 age85*grip9 0;
ESTIMATE 'Grip Strength Slope at Age = 90' grip9 1 age85*grip9 5;
```

**The manual calculations to get predicted values for “fake people” with which to plot the age\*grip interaction are requested via the ESTIMATE statements below.**

```
* If you are not using fake people, you have to write these to create predicted outcomes;
ESTIMATE 'Cognition at Grip = 12 Age = 80' intercept 1 age85 -5 grip9 3 age85*grip9 -15;
ESTIMATE 'Cognition at Grip = 12 Age = 85' intercept 1 age85 0 grip9 3 age85*grip9 0;
ESTIMATE 'Cognition at Grip = 12 Age = 90' intercept 1 age85 5 grip9 3 age85*grip9 15;
ESTIMATE 'Cognition at Grip = 9 Age = 80' intercept 1 age85 -5 grip9 0 age85*grip9 0;
ESTIMATE 'Cognition at Grip = 9 Age = 85' intercept 1 age85 0 grip9 0 age85*grip9 0;
ESTIMATE 'Cognition at Grip = 9 Age = 90' intercept 1 age85 5 grip9 0 age85*grip9 0;
ESTIMATE 'Cognition at Grip = 6 Age = 80' intercept 1 age85 -5 grip9 -3 age85*grip9 15;
ESTIMATE 'Cognition at Grip = 6 Age = 85' intercept 1 age85 0 grip9 -3 age85*grip9 0;
ESTIMATE 'Cognition at Grip = 6 Age = 90' intercept 1 age85 5 grip9 -3 age85*grip9 -15;
RUN; TITLE1; TITLE2;
```

Number of Observations	
Number of Observations Read	559
Number of Observations Used	550
Number of Observations Not Used	9

Note that it did not use our 9 fake people to estimate the model (because they don't have outcomes).

Covariance Parameter Estimates			
Cov Parm	Estimate	Standard Error	Z Value
Residual	86.7615	5.2655	16.48

This is the amount of residual variance remaining after adding the new age\*grip interaction.

Solution for Fixed Effects (values needed for regions of significance in bold)

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	29.4078	0.6949	543	42.32	<.0001
age85	<b>-0.3340</b>	0.1204	543	-2.77	0.0057
grip9	<b>0.6194</b>	0.1487	543	4.16	<.0001
sexMW	-3.4556	0.8873	543	-3.89	0.0001
demNF	-5.9225	1.0136	543	-5.84	<.0001
demNC	-16.3004	1.5125	543	-10.78	<.0001
age85*grip9	<b>0.1230</b>	0.04054	543	3.03	0.0025

**Interpret these fixed effects:**

Intercept  $\beta_0 =$

Simple main effect of Age  $\beta_1 =$

Simple main effect of Grip Strength  $\beta_2 =$

Interpret Age by Grip Strength  $\beta_6 \rightarrow$  Age as Simple Effect, Grip as Moderator:

Interpret Age by Grip Strength  $\beta_6 \rightarrow$  Grip as Simple Effect, Age as Moderator:

Label	Estimates		DF	t Value	Pr >  t
	Estimate	Standard Error			
Age Slope at Grip Strength = 6	-0.7030	0.1534	543	-4.58	<.0001
Age Slope at Grip Strength = 9	-0.3340	0.1204	543	-2.77	0.0057
Age Slope at Grip Strength = 12	0.03509	0.1872	543	0.19	0.8513
Grip Strength Slope at Age = 80	0.004326	0.2473	543	0.02	0.9861
Grip Strength Slope at Age = 85	0.6194	0.1487	543	4.16	<.0001
Grip Strength Slope at Age = 90	1.2345	0.2554	543	4.83	<.0001
Cognition at Grip = 12 Age = 80	31.0906	1.0924	543	28.46	<.0001
Cognition at Grip = 12 Age = 85	31.2661	0.7053	543	44.33	<.0001
Cognition at Grip = 12 Age = 90	31.4415	1.2462	543	25.23	<.0001
Cognition at Grip = 9 Age = 80	31.0776	0.9168	543	33.90	<.0001
Cognition at Grip = 9 Age = 85	29.4078	0.6949	543	42.32	<.0001
Cognition at Grip = 9 Age = 90	27.7380	0.9217	543	30.09	<.0001
Cognition at Grip = 6 Age = 80	31.0646	1.2605	543	24.65	<.0001
Cognition at Grip = 6 Age = 85	27.5495	0.9309	543	29.60	<.0001
Cognition at Grip = 6 Age = 90	24.0345	1.1491	543	20.92	<.0001


```
* Subset predicted outcomes data to fake people;
DATA work.PredAgeGrip; SET work.PredAgeGrip; WHERE PersonID=-99; RUN;
TITLE9 'Predicted Outcomes for Fake People';
PROC PRINT NOOBS DATA=work.PredAgeGrip; VAR age85 grip9 sexMW demNF demNC pred; RUN;
```

Predicted Outcomes for Fake People

age85	grip9	sexmw	dem NF	dem NC	Pred
-5	3	0	0	0	31.0906
0	3	0	0	0	31.2661
5	3	0	0	0	31.4415
-5	0	0	0	0	31.0776
0	0	0	0	0	29.4078
5	0	0	0	0	27.7380
-5	-3	0	0	0	31.0646
0	-3	0	0	0	27.5495
5	-3	0	0	0	24.0345

This printout of the fake cases only is from the "PlotAgeGrip" dataset in SAS (as given by the OUTPM= option) in which the "pred" column is named by default.

From plotting predicted outcomes in excel:



**From adding COVB in SAS MODEL statement (bolded values needed for %regions macro):**

Covariance Matrix for Fixed Effects								
Row	Effect	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	Intercept	0.4829	0.000454	-0.03075	-0.4507	-0.1820	-0.2263	0.001916
2	age85	0.000454	<b>0.01449</b>	0.003317	0.005024	-0.00413	-0.00115	0.000959
3	grip9	-0.03075	0.003317	<b>0.02212</b>	0.05374	-0.01339	-0.00030	0.000203
4	sexMW	-0.4507	0.005024	0.05374	0.7873	-0.07102	0.02371	0.002695
5	demNF	-0.1820	-0.00413	-0.01339	-0.07102	1.0274	0.2129	-0.00268
6	demNC	-0.2263	-0.00115	-0.00030	0.02371	0.2129	2.2878	0.002396
7	age85*grip9	0.001916	0.000959	0.000203	0.002695	-0.00268	0.002396	<b>0.001643</b>

**See excel sheet for calculations, as provided by the SAS macro %Regions below:**

```
* Call macro for regions of significance for main effects of interaction;
%Regions(FixData=FixAgeGrip, CovBData=CovBAgeGrip, Pred=grip9, Mod=age85,
        ModCenter=85, Interact=age85*grip9, Order=6);
```

**Regions of significance for age85\*grip9 interaction:**

The effect of grip9 will be significant at centered values of age85 BELOW the lower bound and ABOVE the upper bound, which translate to these uncentered lower and upper bounds.

Centered Lower	Centered Upper	Uncentered Lower	Uncentered Upper
-14.8174	-2.28519	70.1826	82.7148

So the effect of grip strength will be significantly negative below age = 70.19 years, nonsignificant between age = 70.19 and 82.71 years, and significantly positive after age = 82.71 years.

```
* Call macro for regions of significance for main effects of interaction;
%Regions(FixData=FixAgeGrip, CovBData=CovBAgeGrip, Pred=age85, Mod=grip9,
        ModCenter=9, Interact=age85*grip9, Order=6);
```

**Regions of significance for age85\*grip9 interaction:**

The effect of age85 will be significant at centered values of grip9 BELOW the lower bound and ABOVE the upper bound, which translate to these uncentered lower and upper bounds.

Centered Lower	Centered Upper	Uncentered Lower	Uncentered Upper
0.66541	9.52041	9.66541	18.5204

Similarly, the effect of age will be significantly negative below grip strength = 9.66 pounds, nonsignificant between 9.66 and 18.48 pounds, and significantly positive after 18.48 pounds.

```
* Call macro to calculate R2 for overall model;
%ModelR2(CovBase=CovEmpty, CovFewer=CovDem, CovMore=CovAgeGrip);
```

R2 (% Reduction) Overall and for CovDem vs. CovAgeGrip

Name	CovParm	Estimate	StdErr	ZValue	ProbZ	R2_from_Base	R2_Increment
CovEmpty	Residual	120.76	7.2887	16.57	<.0001	-0.00000	.
CovDem	Residual	88.0709	5.3401	16.49	<.0001	0.27069	.
CovAgeGrip	Residual	86.7615	5.2655	16.48	<.0001	0.28153	0.010843

**MIXED Syntax and Output for Equation 2.18,**

**adding sexMW\*demNC, sexMW\*demNF, age85\*sexMW, grip9\*sexMW, age85\*grip9\*sexMW:**

**Syntax for creating fake people to show three-way interaction in SAS:**

```
* Creating 'fake people' to show age*grip*sex interaction;
* Each row is a fake person for which to create a predicted outcome;
DATA work.FakeAgeGripSex; INPUT personID grip9 age85 sexMW demNF demNC;
DATALINES;
```

-99	3	-5	0	0	0
-99	3	0	0	0	0
-99	3	5	0	0	0
-99	0	-5	0	0	0
-99	0	0	0	0	0
-99	0	5	0	0	0
-99	-3	-5	0	0	0
-99	-3	0	0	0	0
-99	-3	5	0	0	0
-99	3	-5	1	0	0
-99	3	0	1	0	0
-99	3	5	1	0	0
-99	0	-5	1	0	0
-99	0	0	1	0	0
-99	0	5	1	0	0
-99	-3	-5	1	0	0
-99	-3	0	1	0	0
-99	-3	5	1	0	0

Note that the variables that are not part of the age\*grip\*sex interaction are held constant at 0, but they must be included for predicted outcomes to be created.

The dataset "PlotAgeGripSex" will be used in estimating this model instead of the original Chapter2 dataset.

```
; RUN; * Merge with real data;
DATA work.PlotAgeGripSex; MERGE work.Chapter2 work.FakeAgeGripSex; BY personID; RUN;
```

$$\begin{aligned} \text{Cognition}_i &= \beta_0 + \beta_1 (\text{Age}_i - 85) + \beta_2 (\text{Grip}_i - 9) + \beta_3 (\text{SexMW}_i) + \beta_4 (\text{DemNF}_i) + \beta_5 (\text{DemNC}_i) \\ &+ \beta_6 (\text{Age}_i - 85)(\text{Grip}_i - 9) + \beta_7 (\text{SexMW}_i)(\text{DemNF}_i) + \beta_8 (\text{SexMW}_i)(\text{DemNC}_i) \\ &+ \beta_9 (\text{Age}_i - 85)(\text{SexMW}_i) + \beta_{10} (\text{Grip}_i - 9)(\text{SexMW}_i) \\ &+ \beta_{11} (\text{Age}_i - 85)(\text{Grip}_i - 9)(\text{SexMW}_i) + e_i \end{aligned}$$

**Simple main effects** come from:  
 what it is (main effect)  
 + what modifies it (2-ways + 3-ways)

**SAS Model Syntax and Output:**

```
TITLE1 'Eq 2.18: Adding Sex by Age by Grip Three-Way Interaction';
TITLE2 'Age 0=85 Grip 0=9 Sex 0=Men Dementia 0=None';
* Estimate model on data with fake people to make predictions;
PROC MIXED DATA=work.PlotAgeGripSex COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
MODEL cognition = age85 grip9 sexmw demNF demNC
age85*grip9 sexmw*demNF sexmw*demNC
age85*sexMWgrip9*sexMW age85*grip9*sexMW
/ SOLUTION DDFM=BW OUTPM=Pred3AgeSexGrip;
ODS OUTPUT CovParms=Cov3AgeSexGrip;
```

**Simple main effect of age:**

$$\begin{aligned} &\beta_1(\text{Age}_i - 85) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) + \beta_9(\text{Age}_i - 85)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i - 85)(\text{Grip}_i - 9)(\text{SexMW}_i) \\ &= (\text{Age}_i - 85) * [\beta_1 + \beta_6(\text{Grip}_i - 9) + \beta_9(\text{SexMW}_i) + \beta_{11}(\text{Grip}_i - 9)(\text{SexMW}_i)] \end{aligned}$$

```
* Request simple effects of Age;
ESTIMATE "Age for Grip=6, Men" age85 1 age85*grip9 -3 age85*sexMW 0 age85*grip9*sexMW 0;
ESTIMATE "Age for Grip=9, Men" age85 1 age85*grip9 0 age85*sexMW 0 age85*grip9*sexMW 0;
ESTIMATE "Age for Grip=12, Men" age85 1 age85*grip9 3 age85*sexMW 0 age85*grip9*sexMW 0;
ESTIMATE "Age for Grip=6, Women" age85 1 age85*grip9 -3 age85*sexMW 1 age85*grip9*sexMW -3;
ESTIMATE "Age for Grip=9, Women" age85 1 age85*grip9 0 age85*sexMW 1 age85*grip9*sexMW 0;
ESTIMATE "Age for Grip=12, Women" age85 1 age85*grip9 3 age85*sexMW 1 age85*grip9*sexMW 3;
```

**Simple main effect of grip strength:**

$$\beta_2(\text{Grip}_i-9) + \beta_6(\text{Age}_i-85)(\text{Grip}_i-9) + \beta_{10}(\text{Grip}_i-9)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i-85)(\text{Grip}_i-9)(\text{SexMW}_i)$$

$$= (\text{Grip}_i-9) * [\beta_2 + \beta_6(\text{Age}_i-85) + \beta_{10}(\text{SexMW}_i) + \beta_{11}(\text{Age}_i-85)(\text{SexMW}_i)]$$

\* Request simple effects of Grip;

```
ESTIMATE "Grip for Age=80, Men"   grip9 1 age85*grip9 -5 grip9*sexMW 0 age85*grip9*sexMW 0;
ESTIMATE "Grip for Age=85, Men"   grip9 1 age85*grip9 0 grip9*sexMW 0 age85*grip9*sexMW 0;
ESTIMATE "Grip for Age=90, Men"   grip9 1 age85*grip9 5 grip9*sexMW 0 age85*grip9*sexMW 0;
ESTIMATE "Grip for Age=80, Women" grip9 1 age85*grip9 -5 grip9*sexMW 1 age85*grip9*sexMW -5;
ESTIMATE "Grip for Age=85, Women" grip9 1 age85*grip9 0 grip9*sexMW 1 age85*grip9*sexMW 0;
ESTIMATE "Grip for Age=90, Women" grip9 1 age85*grip9 5 grip9*sexMW 1 age85*grip9*sexMW 5;
```

**Simple main effect of sex (hold dementia group at 0=none):**

$$\beta_3(\text{SexMW}_i) + \beta_9(\text{Age}_i-85)(\text{SexMW}_i) + \beta_{10}(\text{Grip}_i-9)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i-85)(\text{Grip}_i-9)(\text{SexMW}_i)$$

$$= (\text{SexMW}_i) * [\beta_3 + \beta_9(\text{Age}_i-85) + \beta_{10}(\text{Grip}_i-9) + \beta_{11}(\text{Age}_i-85)(\text{Grip}_i-9)]$$

\* Request simple effects of Sex (hold demgroup 0=None);

```
ESTIMATE "Sex: Age=80, Grip=6"   sexMW 1 age85*sexMW-5 grip9*sexMW -3 age85*grip9*sexMW 15;
ESTIMATE "Sex: Age=85, Grip=6"   sexMW 1 age85*sexMW 0 grip9*sexMW -3 age85*grip9*sexMW 0;
ESTIMATE "Sex: Age=90, Grip=6"   sexMW 1 age85*sexMW 5 grip9*sexMW -3 age85*grip9*sexMW -15;
ESTIMATE "Sex: Age=80, Grip=9"   sexMW 1 age85*sexMW-5 grip9*sexMW 0 age85*grip9*sexMW 0;
ESTIMATE "Sex: Age=85, Grip=9"   sexMW 1 age85*sexMW 0 grip9*sexMW 0 age85*grip9*sexMW 0;
ESTIMATE "Sex: Age=90, Grip=9"   sexMW 1 age85*sexMW 5 grip9*sexMW 0 age85*grip9*sexMW 0;
ESTIMATE "Sex: Age=80, Grip=12"  sexMW 1 age85*sexMW-5 grip9*sexMW 3 age85*grip9*sexMW -15;
ESTIMATE "Sex: Age=85, Grip=12"  sexMW 1 age85*sexMW 0 grip9*sexMW 3 age85*grip9*sexMW 0;
ESTIMATE "Sex: Age=90, Grip=12"  sexMW 1 age85*sexMW 5 grip9*sexMW 3 age85*grip9*sexMW 15;
```

**Simple two-way interaction of age\*grip:**  $\beta_6(\text{Age}_i-85)(\text{Grip}_i-9) + \beta_{11}(\text{Age}_i-85)(\text{Grip}_i-9)(\text{SexMW}_i)$

\* Request simple two-way interactions of age\*grip;

```
ESTIMATE "Age by Grip: Men"      age85*grip9 1 age85*grip9*sexMW 0;
ESTIMATE "Age by Grip: Women"   age85*grip9 1 age85*grip9*sexMW 1;
```

**Simple two-way interaction of age\*sex:**  $\beta_9(\text{Age}_i-85)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i-85)(\text{Grip}_i-9)(\text{SexMW}_i)$

\* Request simple two-way interactions of age\*sex;

```
ESTIMATE "Sex by Age: Grip=6"    age85*sexMW 1 age85*grip9*sexMW -3;
ESTIMATE "Sex by Age: Grip=9"    age85*sexMW 1 age85*grip9*sexMW 0;
ESTIMATE "Sex by Age: Grip=12"   age85*sexMW 1 age85*grip9*sexMW 3;
```

**Simple two-way interaction of grip\*sex:**  $\beta_{10}(\text{Grip}_i-9)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i-85)(\text{Grip}_i-9)(\text{SexMW}_i)$

\* Request simple two-way interactions of grip\*sex;

```
ESTIMATE "Sex by Grip: Age=80"   grip9*sexMW 1 age85*grip9*sexMW -5;
ESTIMATE "Sex by Grip: Age=85"   grip9*sexMW 1 age85*grip9*sexMW 0;
ESTIMATE "Sex by Grip: Age=90"   grip9*sexMW 1 age85*grip9*sexMW 5;
RUN; TITLE1; TITLE2;
```

Number of Observations	
Number of Observations Read	568
Number of Observations Used	550
Number of Observations Not Used	18

Note that it did not use our 18 fake people to estimate the model (because they don't have outcomes).

Covariance Parameter Estimates				
		Standard	Z	
Cov Parm	Estimate	Error	Value	Pr > Z
Residual	85.9433	5.2401	16.40	<.0001

This is the amount of residual variance remaining after adding the new interactions.

Solution for Fixed Effects  
Standard

Effect	Estimate	Error	DF	t Value	Pr >  t
Intercept	28.9558	0.7972	538	36.32	<.0001
age85	-0.4983	0.2063	538	-2.42	0.0160
grip9	0.7393	0.2364	538	3.13	0.0019
sexMW	-2.9735	1.0364	538	-2.87	0.0043
demNF	-6.1654	1.6379	538	-3.76	0.0002
demNC	-11.7837	2.2479	538	-5.24	<.0001
age85*grip9	0.2300	0.07595	538	3.03	0.0026
sexMW*demNF	0.3969	2.0796	538	0.19	0.8487
sexMW*demNC	-8.1457	3.0283	538	-2.69	0.0074
age85*sexMW	0.09507	0.2742	538	0.35	0.7289
grip9*sexMW	-0.2001	0.3046	538	-0.66	0.5115
age85*grip9*sexMW	-0.1583	0.09532	538	-1.66	0.0974

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Age for Grip=6, Men	-1.1882	0.3659	538	-3.25	0.0012
Age for Grip=9, Men	-0.4983	0.2063	538	-2.42	0.0160
Age for Grip=12, Men	0.1916	0.2346	538	0.82	0.4146
Age for Grip=6, Women	-0.6183	0.1694	538	-3.65	0.0003
Age for Grip=9, Women	-0.4032	0.1806	538	-2.23	0.0260
Age for Grip=12, Women	-0.1882	0.3103	538	-0.61	0.5446
Grip for Age=80, Men	-0.4105	0.4368	538	-0.94	0.3478
Grip for Age=85, Men	0.7393	0.2364	538	3.13	0.0019
Grip for Age=90, Men	1.8891	0.4576	538	4.13	<.0001
Grip for Age=80, Women	0.1808	0.3294	538	0.55	0.5834
Grip for Age=85, Women	0.5392	0.1920	538	2.81	0.0052
Grip for Age=90, Women	0.8976	0.3622	538	2.48	0.0135
Sex: Age=80, Grip=6	-5.2227	2.5462	538	-2.05	0.0407
Sex: Age=85, Grip=6	-2.3733	1.4657	538	-1.62	0.1060
Sex: Age=90, Grip=6	0.4761	2.4373	538	0.20	0.8452
Sex: Age=80, Grip=9	-3.4489	1.6404	538	-2.10	0.0360
Sex: Age=85, Grip=9	-2.9735	1.0364	538	-2.87	0.0043
Sex: Age=90, Grip=9	-2.4981	1.7931	538	-1.39	0.1641
Sex: Age=80, Grip=12	-1.6750	2.0704	538	-0.81	0.4188
Sex: Age=85, Grip=12	-3.5737	1.2921	538	-2.77	0.0059
Sex: Age=90, Grip=12	-5.4724	2.5730	538	-2.13	0.0339
Age by Grip: Men	0.2300	0.07595	538	3.03	0.0026
Age by Grip: Women	0.07169	0.05761	538	1.24	0.2139
Age by Sex: Grip=6	0.5699	0.4032	538	1.41	0.1581
Age by Sex: Grip=9	0.09507	0.2742	538	0.35	0.7289
Age by Sex: Grip=12	-0.3797	0.3890	538	-0.98	0.3295
Grip by Sex: Age=80	0.5913	0.5471	538	1.08	0.2803
Grip by Sex: Age=85	-0.2001	0.3046	538	-0.66	0.5115
Grip by Sex: Age=90	-0.9914	0.5836	538	-1.70	0.0899

```
* Subset predicted outcomes data to fake people;
DATA work.Pred3AgeSexGrip; SET work.Pred3AgeSexGrip; WHERE PersonID=-99; RUN;
TITLE9 'Predicted Outcomes for Fake People';
PROC PRINT NOOBS DATA=work.Pred3AgeSexGrip; VAR age85 grip9 sexmw demNF demNC pred; RUN;
```

Predicted Outcomes for Fake People

age85	grip9	sexmw	dem		Pred
			NF	NC	
-5	3	0	0	0	30.2157
0	3	0	0	0	31.1737
5	3	0	0	0	32.1316
-5	0	0	0	0	31.4473
0	0	0	0	0	28.9558
5	0	0	0	0	26.4644
-5	-3	0	0	0	32.6788
0	-3	0	0	0	26.7380
5	-3	0	0	0	20.7972
-5	3	1	0	0	28.5407
0	3	1	0	0	27.6000
5	3	1	0	0	26.6592
-5	0	1	0	0	27.9984
0	0	1	0	0	25.9824
5	0	1	0	0	23.9663
-5	-3	1	0	0	27.4561
0	-3	1	0	0	24.3648
5	-3	1	0	0	21.2734

This printout of the fake cases only is from the "PlotAgeGripSex" dataset in SAS (as given by the OUTPM= option) in which the "pred" column is named by default.

**From plotting predicted outcomes in excel:**

